

# Querying data from data.bnf.fr with SPARQL

## 1 Documentation of data.bnf.fr

Go to the BNF Sparql service <http://data.bnf.fr/sparql> . Test some pre-loaded queries

## 2 Querying data.bnf.fr

**A)** Look for a person called "Jean de La Fontaine". A person's name is expressed with the property " foaf:name " .

You can consult the RDF data of his card to help you.

Jean de la Fontaine = <http://data.bnf.fr/ark:/12148/cb11910267w#about>

Solution :

```
SELECT ?p
WHERE {
?p a foaf:Person .
?p foaf:name "Jean de La Fontaine" .
}
```

To disambiguate the answers, we can display date of birth and death :

```
SELECT ?p ?birth ?death
WHERE {
?p a foaf:Person .
?p foaf:name "Jean de La Fontaine" .
?p bio:birth ?birth .
?p bio:death ?death .
}
```

**B)** List of literary works written by Jean de La Fontaine.

```
PREFIX dcterms: <http://purl.org/dc/terms/>
SELECT ?oeuvre
WHERE {
?oeuvre dcterms:creator
<http://data.bnf.fr/ark:/12148/cb11910267w#about> .
}
```

**C)** And their title ?

```
PREFIX dcterms: <http://purl.org/dc/terms/>
SELECT ?oeuvre ?title
WHERE {
?oeuvre dcterms:creator
<http://data.bnf.fr/ark:/12148/cb11910267w#about> .
?oeuvre dcterms:title ?title .
}
```

**D)** Enrich the query to find editions of literary works written by Jean de La Fontaine.

```
PREFIX dcterms: <http://purl.org/dc/terms/>
```

```

PREFIX rdarelationshps: <http://rdvocab.info/RDARelationshipsWEMI/>
SELECT ?oeuvre ?titre ?edition
WHERE {
?oeuvre dcterms:creator
<http://data.bnf.fr/ark:/12148/cb11910267w#about> .
?oeuvre dcterms:title ?titre .
?edition rdarelationshps:workManifested ?oeuvre .
}

```

**E)** And their title ?

```

PREFIX dcterms: <http://purl.org/dc/terms/>
PREFIX rdarelationshps: <http://rdvocab.info/RDARelationshipsWEMI/>
SELECT ?oeuvre ?titreOeuvre ?edition ?titreEdition
WHERE {
?oeuvre dcterms:creator
<http://data.bnf.fr/ark:/12148/cb11910267w#about> .
?oeuvre dcterms:title ?titreOeuvre .
?edition rdarelationshps:workManifested ?oeuvre .
?edition dcterms:title ?titreEdition
}

```

**F)** And their ISBN ?

How to get the ISBN of these editions? (all editions do not have a ISBN)

```

PREFIX dcterms: <http://purl.org/dc/terms/>
PREFIX rdarelationshps: <http://rdvocab.info/RDARelationshipsWEMI/>
PREFIX bnf-onto: <http://data.bnf.fr/ontology/bnf-onto/>
SELECT ?oeuvre ?titreOeuvre ?edition ?titreEdition ?isbn
WHERE {
?oeuvre dcterms:creator
<http://data.bnf.fr/ark:/12148/cb11910267w#about> .
?oeuvre dcterms:title ?titreOeuvre .
?edition rdarelationshps:workManifested ?oeuvre .
?edition dcterms:title ?titreEdition .
OPTIONAL { ?edition bnf-onto:isbn ?isbn . }
}

```

# Querying data from DBpedia with SPARQL

Go to the URI DBpedia of Film <http://dbpedia.org/ontology/Film> and discover the different properties.

Go to the DbPedia Sparql service <http://dbpedia.org/sparql/> and write a query to display the list of film titles.

```
SELECT ?titre
WHERE {
  ?film a <http://dbpedia.org/ontology/Film> .
  ?film rdfs:label ?titre }
LIMIT 20
```

Ou bien

```
PREFIX dbo:<http://dbpedia.org/ontology/>
SELECT ?titre
WHERE {
  ?film a dbo:Film .
  ?film rdfs:label ?titre }
LIMIT 20
```

And the abstracts

```
PREFIX dbo:<http://dbpedia.org/ontology/>
SELECT ?titre ?abstract
WHERE {
  ?film a <http://dbpedia.org/ontology/Film> .
  ?film rdfs:label ?titre .
  ?film dbo:abstract ?abstract}
LIMIT 20
```

And the wiki pages

```
PREFIX dbo:<http://dbpedia.org/ontology/>
SELECT ?titre ?abstract ?wiki
WHERE {
  ?film a <http://dbpedia.org/ontology/Film> .
  ?film rdfs:label ?titre .
  ?film dbo:abstract ?abstract.
  ?film foaf:isPrimaryTopicOf ?wiki}
LIMIT 20
```